Computational Optimization for Economists

Sven Leyffer, Jorge Moré, and Todd Munson
Mathematics and Computer Division
Argonne National Laboratory
{leyffer,more,tmunson}@mcs.anl.gov

Institute for Computational Economics
Argonne National Laboratory
July 17-21, 2006
Computational Optimization Overview

1. Introducion to Optimization [Moré]
2. Continuous Optimization in AMPL [Munson]
3. Optimization Software [Leyffer]
4. Complementarity & Games [Munson]
Part I

Introduction, Applications, and Formulations
Outline

• Software
  • Views of optimization
  • Characteristic of optimization software
  • Case studies in optimization software

• Environments
  • Modeling Languages: AMPL, GAMS
  • Solving optimization problems
  • Automatic differentiation

• Tools
  • Benchmarking
  • Performance profiles
  • Scale invariance
Nonlinearly Constrained Optimization

\[
\min \left\{ f(x) : x_l \leq x \leq x_u, \ c_l \leq c(x) \leq c_u \right\}
\]

- Objective function is defined by \( f : \mathbb{R}^n \mapsto \mathbb{R} \)
- Constraints are defined by \( c : \mathbb{R}^n \mapsto \mathbb{R}^m \).
- Bounds \( x_l \leq x \leq x_u \) on the variables \( x \in \mathbb{R}^n \).

- First-order algorithms require the gradient
  \[
  \nabla f(x) = (\partial_i f(x)), \quad \nabla c_1(x), \ldots, \nabla c_m(x)
  \]
- Second order algorithms require the Hessians
  \[
  \nabla^2 f(x) = (\partial_{i,j} f(x)), \quad \nabla^2 c_1(x), \ldots, \nabla^2 c_m(x)
  \]
Part II

Continuous Optimization in AMPL
Modeling Languages

- Portable language for optimization problems
  - Algebraic description
  - Models easily modified and solved
  - Large problems can be processed
  - Programming language features

- Many available optimization algorithms
  - No need to compile C/FORTRAN code
  - Derivatives automatically calculated
  - Algorithms specific options can be set

- Communication with other tools
  - Relational databases and spreadsheets
  - MATLAB interface for function evaluations

- Excellent documentation

- Large user communities
Model Declaration

- **Sets**
  - Unordered, ordered, and circular sets
  - Cross products and point to set mappings
  - Set manipulation

- **Parameters and variables**
  - Initial and default values
  - Lower and upper bounds
  - Check statements
  - Defined variables

- **Objective function and constraints**
  - Equality, inequality, and range constraints
  - Complementarity constraints
  - Multiple objectives

- **Problem statement**
Data and Commands

- Data declaration
  - Set definitions
    - Explicit list of elements
    - Implicit list in parameter statements
  - Parameter definitions
    - Tables and transposed tables
    - Higher dimensional parameters

- Execution commands
  - Load model and data
  - Select problem, algorithm, and options
  - Solve the instance
  - Output results

- Other operations
  - Let and fix statements
  - Conditionals and loop constructs
  - Execution of external programs
Model Formulation

- Economy with $n$ agents and $m$ commodities
  - $e \in \mathbb{R}^{n \times m}$ are the endowments
  - $\alpha \in \mathbb{R}^{n \times m}$ and $\beta \in \mathbb{R}^{n \times m}$ are the utility parameters
  - $\lambda \in \mathbb{R}^n$ are the social weights

- Social planning problem

$$\max_{x \geq 0} \sum_{i=1}^{n} \lambda_i \left( \sum_{k=1}^{m} \frac{\alpha_{i,k} (1 + x_{i,k})^{1-\beta_{i,k}}}{1 - \beta_{i,k}} \right)$$

subject to

$$\sum_{i=1}^{n} x_{i,k} \leq \sum_{i=1}^{n} e_{i,k} \quad \forall k = 1, \ldots, m$$
Model: social1.mod

param n > 0, integer;  # Agents
param m > 0, integer;  # Commodities

param e {1..n, 1..m} >= 0, default 1;  # Endowment

param lambda {1..n} > 0;  # Social weights
param alpha {1..n, 1..m} > 0;  # Utility parameters
param beta {1..n, 1..m} > 0;

var x{1..n, 1..m} >= 0;  # Consumption
var u{i in 1..n} =  # Utility
    sum {k in 1..m} alpha[i,k] * (1 + x[i,k])^(1 - beta[i,k]) / (1 - beta[i,k]);

maximize welfare:
    sum {i in 1..n} lambda[i] * u[i];

subject to
    consumption {k in 1..m}:
        sum {i in 1..n} x[i,k] <= sum {i in 1..n} e[i,k];
### Data: social1.dat

```plaintext
param n := 3;          # Agents
param m := 4;          # Commodities

param alpha : 1 2 3 4 :=
   1 1 1 1 1
   2 1 2 3 4
   3 2 1 1 5;

param beta (tr) : 1 2 3 :=
   1 1.5 2 0.6
   2 1.6 3 0.7
   3 1.7 2 2.0
   4 1.8 2 2.5;

param : lambda :=
   1 1
   2 1
   3 1;
```
Commands: social1.cmd

# Load model and data
model social1.mod;
data social1.dat;

# Specify solver and options
option solver "minos";
option minos_options "outlev=1";

# Solve the instance
solve;

# Output results
display x;
printf {i in 1..n} "%2d: % 5.4e\n", i, u[i];
Output

ampl: include social1.cmd;
MINOS 5.5: outlev=1
MINOS 5.5: optimal solution found.
25 iterations, objective 2.252422003
Nonlin evals: obj = 44, grad = 43.
x :=
  1  1  0.0811471
  1  2  0.574164
  1  3  0.703454
  1  4  0.267241
  2  1  0.060263
  2  2  0.604858
  2  3  1.7239
  2  4  1.47516
  3  1  2.85859
  3  2  1.82098
  3  3  0.572645
  3  4  1.2576
;

  1:  -5.2111e+00
  2:  -4.0488e+00
  3:  1.1512e+01
ampl: quit;
Model: social2.mod

set AGENTS;                     # Agents
set COMMODITIES;                # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param lambda {AGENTS} > 0;      # Social weights
param alpha {AGENTS, COMMODITIES} > 0; # Utility parameters
param beta {AGENTS, COMMODITIES} > 0;

param gamma {i in AGENTS, k in COMMODITIES} := 1 - beta[i,k];

var x{AGENTS, COMMODITIES} >= 0; # Consumption
var u{i in AGENTS} =                        # Utility
    sum {k in COMMODITIES} alpha[i,k] * (1 + x[i,k])^gamma[i,k] / gamma[i,k];

maximize welfare:
    sum {i in AGENTS} lambda[i] * u[i];

subject to
    consumption {k in COMMODITIES}:
        sum {i in AGENTS} x[i,k] <= sum {i in AGENTS} e[i,k];
Data: social2.dat

set COMMODITIES := Books, Cars, Food, Pens;

param: AGENTS : lambda :=
    Jorge 1
    Sven 1
    Todd 1;

param alpha : Books Cars Food Pens :=
    Jorge 1 1 1 1
    Sven 1 2 3 4
    Todd 2 1 1 5;

param beta (tr): Jorge Sven Todd :=
    Books 1.5 2 0.6
    Cars 1.6 3 0.7
    Food 1.7 2 2.0
    Pens 1.8 2 2.5;
Commands: social2.cmd

# Load model and data
model social2.mod;
data social2.dat;

# Specify solver and options
option solver "minos";
option minos_options "outlev=1";

# Solve the instance
solve;

# Output results
display x;
printf {i in AGENTS} "%5s: % 5.4e\n", i, u[i];
Output

ampl: include social2.cmd
MINOS 5.5: outlev=1
MINOS 5.5: optimal solution found.
25 iterations, objective 2.252422003
Nonlin evals: obj = 44, grad = 43.
x :=
  Jorge Books 0.0811471
  Jorge Cars 0.574164
  Jorge Food 0.703454
  Jorge Pens 0.267241
  Sven Books 0.060263
  Sven Cars 0.604858
  Sven Food 1.7239
  Sven Pens 1.47516
  Todd Books 2.85859
  Todd Cars 1.82098
  Todd Food 0.572645
  Todd Pens 1.2576
;

Jorge: -5.2111e+00
Sven: -4.0488e+00
Todd: 1.1512e+01
ampl: quit;
Model Formulation

- Route commodities through a network
  - $\mathcal{N}$ is the set of nodes
  - $\mathcal{A} \subseteq \mathcal{N} \times \mathcal{N}$ is the set of arcs
  - $\mathcal{K}$ is the set of commodities
  - $\alpha$ and $\beta$ are the congestion parameters
  - $b$ denotes the supply and demand

- Multicommodity network flow problem

$$\max_{x \geq 0, f \geq 0} \sum_{(i,j) \in \mathcal{A}} \left( \alpha_{i,j} f_{i,j} + \beta_{i,j} f_{i,j}^4 \right)$$

subject to

$$\sum_{(i,j) \in \mathcal{A}} x_{i,j,k} \leq \sum_{(j,i) \in \mathcal{A}} x_{j,i,k} + b_{i,k} \quad \forall i \in \mathcal{N}, k \in \mathcal{K}$$

$$f_{i,j} = \sum_{k \in \mathcal{K}} x_{i,j,k} \quad \forall (i, j) \in \mathcal{A}$$
Model: network.mod

set NODES; # Nodes in network
set ARCS within NODES cross NODES; # Arcs in network
set COMMODITIES := 1..3; # Commodities

param b {NODES, COMMODITIES} default 0; # Supply/demand
check {k in COMMODITIES}:
   sum {i in NODES} b[i,k] >= 0;

param alpha{ARCS} >= 0; # Linear part
param beta{ARCS} >= 0; # Nonlinear part

var x{ARCS, COMMODITIES} >= 0; # Flow on arcs
var f{(i,j) in ARCS} =
   sum {k in COMMODITIES} x[i,j,k]; # Total flow

minimize time:
   sum {(i,j) in ARCS} (alpha[i,j]*f[i,j] + beta[i,j]*f[i,j]^4);

subject to
   conserve {i in NODES, k in COMMODITIES}:
      sum {(i,j) in ARCS} x[i,j,k] <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];
Data: network.dat

set NODES := 1 2 3 4 5;

param: ARCS : alpha beta =
    1 2 1 0.5
    1 3 1 0.4
    2 3 2 0.7
    2 4 3 0.1
    3 2 1 0.0
    3 4 4 0.5
    4 1 5 0.0
    4 5 2 0.1
    5 2 0 1.0;

let b[1,1] := 7; # Node 1, Commodity 1 supply
let b[4,1] := -7; # Node 4, Commodity 1 demand
let b[2,2] := 3;  # Node 2, Commodity 2 supply
let b[5,2] := -3; # Node 5, Commodity 2 demand
let b[3,3] := 5;  # Node 1, Commodity 3 supply
let b[1,3] := -5; # Node 4, Commodity 3 demand

fix {i in NODES, k in COMMODITIES: (i,i) in ARCS} x[i,i,k] := 0;
# Load model and data
model network.mod;
data network.dat;

# Specify solver and options
option solver "minos";
option minos_options "outlev=1";

# Solve the instance
solve;

# Output results
for {k in COMMODITIES} {
    printf "Commodity: %d\n", k > network.out;
    printf {((i,j) in ARCS: x[i,j,k] > 0)} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > network.out;
    printf "\n" > network.out;
}

Leyffer, Moré, and Munson
Computational Optimization
Output

ampl: include network.cmd;
MINOS 5.5: outlev=1
MINOS 5.5: optimal solution found.
12 iterations, objective 1505.526478
Nonlin evals: obj = 14, grad = 13.
ampl: quit;
Results: network.out

Commodity: 1
1.2 = 3.3775e+00
1.3 = 3.6225e+00
2.4 = 6.4649e+00
3.2 = 3.0874e+00
3.4 = 5.3510e-01

Commodity: 2
2.4 = 3.0000e+00
4.5 = 3.0000e+00

Commodity: 3
3.4 = 5.0000e+00
4.1 = 5.0000e+00
Initial Coordinate Descent: wardrop0.cmd

# Load model and data
model network.mod;
data network.dat;

option solver "minos";
option minos_options "outlev=1";

# Coordinate descent method
fix {(i,j) in ARCS, k in COMMODITIES} x[i,j,k];
drop {i in NODES, k in COMMODITIES} conserve[i,k];

for {iter in 1..100} {
    for {k in COMMODITIES} {
        unfix {(i,j) in ARCS} x[i,j,k];
        restore {i in NODES} conserve[i,k];

        solve {i in NODES} conserve[i,k];

        fix {(i,j) in ARCS} x[i,j,k];
        drop {i in NODES} conserve[i,k];
    }
}

# Output results
for {k in COMMODITIES} {
    printf "\nCommodity: %d\n", k > network.out;
    printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > network.out;
}
Improved Coordinate Descent: wardrop.mod

```plaintext
set NODES;                   # Nodes in network
set ARCS within NODES cross NODES;  # Arcs in network
set COMMODITIES := 1..3;       # Commodities

param b {NODES, COMMODITIES} default 0;        # Supply/demand
param alpha {ARCS} >= 0;                   # Linear part
param beta {ARCS} >= 0;                  # Nonlinear part

var x {ARCS, COMMODITIES} >= 0;        # Flow on arcs
var f {(i,j) in ARCS} =               # Total flow
    sum {k in COMMODITIES} x[i,j,k];

minimize time {k in COMMODITIES}:
    sum {(i,j) in ARCS} (alpha[i,j]*f[i,j] + beta[i,j]*f[i,j]^4);

subject to
    conserve {i in NODES, k in COMMODITIES}:
        sum {(i,j) in ARCS} x[i,j,k] <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];

problem subprob {k in COMMODITIES}: time[k], {i in NODES} conserve[i,k],
    {(i,j) in ARCS} x[i,j,k], f;
```
Improved Coordinate Descent: wardrop1.cmd

# Load model and data
model wardrop.mod;
data wardrop.dat;

# Specify solver and options
option solver "minos";
option minos_options "outlev=1";

# Coordinate descent method
for {iter in 1..100} {
    for {k in COMMODITIES} {
        solve subprob[k];
    }
}

for {k in COMMODITIES} {
    printf "Commodity: %d\n", k > wardrop.out;
    printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > wardrop.out;
    printf "\n" > wardrop.out;
}
Final Coordinate Descent: wardrop2.cmd

# Load model and data
model wardrop.mod;
data wardrop.dat;

# Specify solver and options
option solver "minos";
option minos_options "outlev=1";

# Coordinate descent method
param xold{ARCS, COMMODITIES};
param xnew{ARCS, COMMODITIES};
repeat {
  for {k in COMMODITIES} {
    problem subprob[k];
    let {(i,j) in ARCS} xold[i,j,k] := x[i,j,k];
    solve;
    let {(i,j) in ARCS} xnew[i,j,k] := x[i,j,k];
  }
} until (sum {(i,j) in ARCS, k in COMMODITIES} abs(xold[i,j,k] - xnew[i,j,k]) <= 1e-6);

for {k in COMMODITIES} {
  printf "Commodity: %d\n", k > wardrop.out;
  printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.\%d = % 5.4e\n", i, j, x[i,j,k] > wardrop.out;
  printf "\n" > wardrop.out;
}
Ordered Sets

```
param V, integer; # Number of vertices
param E, integer; # Number of elements

set VERTICES := {1..V}; # Vertex indices
set ELEMENTS := {1..E}; # Element indices
set COORDS := {1..3} ordered; # Spatial coordinates

param T{ELEMENTS, 1..4} in VERTICES; # Tetrahedral elements

var x{VERTICES, COORDS}; # Position of vertices

var norm{e in ELEMENTS} = sum{i in COORDS, j in 1..4}
(x[T[e,j], i] - x[T[e,1], i])^2;

var area{e in ELEMENTS} = sum{i in COORDS}
((x[T[e,3], nextw(i)] - x[T[e,1], nextw(i)]) *
(x[T[e,4], prevw(i)] - x[T[e,1], prevw(i)]) -
(x[T[e,3], prevw(i)] - x[T[e,1], prevw(i)]) *
(x[T[e,4], nextw(i)] - x[T[e,1], nextw(i)]));

minimize f: sum {e in ELEMENTS} norm[e] / max(area[e], 0) ^ (2 / 3);
```
Circular Sets

param V, integer;                  # Number of vertices
param E, integer;                  # Number of elements

set VERTICES := {1..V};            # Vertex indices
set ELEMENTS := {1..E};            # Element indices
set COORDS := {1..3} circular;     # Spatial coordinates

param T{ELEMENTS, 1..4} in VERTICES; # Tetrahedral elements

var x{VERTICES, COORDS};          # Position of vertices

var norm{e in ELEMENTS} = sum{i in COORDS, j in 1..4}
    (x[T[e,j], i] - x[T[e,1], i])^2;

var area{e in ELEMENTS} = sum{i in COORDS}
    (x[T[e,2], i] - x[T[e,1], i])
    * ((x[T[e,3], next(i)] - x[T[e,1], next(i)])
    * (x[T[e,4], prev(i)] - x[T[e,1], prev(i)])
    - (x[T[e,3], prev(i)] - x[T[e,1], prev(i)])
    * (x[T[e,4], next(i)] - x[T[e,1], next(i)]));

minimize f: sum {e in ELEMENTS} norm[e] / max(area[e], 0) ^ (2 / 3);
Part III

Optimization Software
Generic Nonlinear Optimization Problem

Nonlinear Programming (NLP) problem

\[
\begin{aligned}
\text{minimize} & \quad f(x) \quad \text{objective} \\
\text{subject to} & \quad c(x) = 0 \quad \text{constraints} \\
& \quad x \geq 0 \quad \text{variables}
\end{aligned}
\]

- \(f : \mathbb{R}^n \rightarrow \mathbb{R}, \ c : \mathbb{R}^n \rightarrow \mathbb{R}^m\) smooth (typically \(C^2\))
- \(x \in \mathbb{R}^n\) finite dimensional (may be large)
- more general \(l \leq c(x) \leq u\) possible
Optimality Conditions for NLP

Constraint qualification (CQ)
Linearizations of \( c(x) = 0 \) characterize all feasible perturbations
\( \Rightarrow \) rules out cusps etc.

\( x^* \) local minimizer \& CQ holds \( \Rightarrow \exists \) multipliers \( y^* , z^* \):

\[
\nabla f(x^*) - \nabla c(x^*)^T y^* - z^* = 0 \\
c(x^*) = 0 \\
X^* z^* = 0 \\
x^* \geq 0, \ z^* \geq 0
\]

where \( X^* = \text{diag}(x^*) \), thus \( X^* z^* = 0 \) \( \iff \ x_i^* z_i^* = 0 \)

Lagrangian: \( \mathcal{L}(x, y, z) := f(x) - y^T c(x) - z^T x \)
Optimality Conditions for NLP

Objective gradient is linear combination of constraint gradients

$$g(x) = A(x)y, \quad \text{where } g(x) := \nabla f(x), \ A(x) := \nabla c(x)^T$$
Newton’s Method for Nonlinear Equations

Solve \( F(x) = 0 \):

Get approx. \( x_{k+1} \) of solution of \( F(x) = 0 \) by solving linear model about \( x_k \):

\[
F(x_k) + \nabla F(x_k)^T (x - x_k) = 0
\]

for \( k = 0, 1, \ldots \)

**Theorem:** If \( F \in C^2 \), and \( \nabla F(x^*) \) nonsingular, then Newton converges quadratically near \( x^* \).
Newton’s Method for Nonlinear Equations

Next: two classes of methods based on Newton ...
Sequential Quadratic Programming (SQP)

Consider equality constrained NLP

\[
\begin{align*}
\min_x f(x) \quad \text{subject to } c(x) &= 0
\end{align*}
\]

Optimality conditions:

\[
\nabla f(x) - \nabla c(x)^T y = 0 \quad \text{and} \quad c(x) = 0
\]

... system of nonlinear equations: \( F(w) = 0 \) for \( w = (x, y) \).

... solve using Newton’s method
Sequential Quadratic Programming (SQP)

Nonlinear system of equations (KKT conditions)

\[ \nabla f(x) - \nabla c(x)^T y = 0 \quad \text{and} \quad c(x) = 0 \]

Apply Newton’s method from \( w_k = (x_k, y_k) \) \( \ldots \)

\[ H_k = \nabla^2 \mathcal{L}(x_k, y_k) \]

\[ \begin{bmatrix} H_k & -A_k \\ A_k^T & 0 \end{bmatrix} \begin{pmatrix} s_x \\ s_y \end{pmatrix} = -\begin{pmatrix} \nabla_x \mathcal{L}(x_k, y_k) \\ c_k \end{pmatrix} \]

\[ \ldots \text{set} \ (x_{k+1}, y_{k+1}) = (x_k + s_x, y_k + s_y) \quad \ldots \]

\[ A^k = \nabla c(x_k)^T \]

\[ \ldots \text{solve for} \ y_{k+1} = y_k + s_y \quad \text{directly instead:} \]

\[ \begin{bmatrix} H_k & -A_k \\ A_k^T & 0 \end{bmatrix} \begin{pmatrix} s \\ y_{k+1} \end{pmatrix} = -\begin{pmatrix} \nabla f_k \\ c_k \end{pmatrix} \]

\[ \ldots \text{set} \ (x_{k+1}, y_{k+1}) = (x_k + s, y_{k+1}) \]
Sequential Quadratic Programming (SQP)

Newton’s Method for KKT conditions leads to:

\[
\begin{bmatrix}
H_k & -A_k \\
A^T_k & 0
\end{bmatrix}
\begin{pmatrix}
s \\
y_{k+1}
\end{pmatrix}
= 
-\begin{pmatrix}
\nabla f_k \\
c_k
\end{pmatrix}
\]

... are optimality conditions of QP

\[
\begin{aligned}
&\text{minimize} & & \nabla f_k^T s + \frac{1}{2} s^T H_k s \\
&\text{subject to} & & c_k + A_k^T s = 0
\end{aligned}
\]

... hence Sequential Quadratic Programming
Sequential Quadratic Programming (SQP)

SQP for inequality constrained NLP:

\[
\begin{aligned}
&\text{minimize } f(x) \quad \text{subject to } c(x) = 0 \quad \& \quad x \geq 0 \\
\end{aligned}
\]

REPEAT

1. Solve QP for \((s, y_{k+1}, z_{k+1})\)

\[
\begin{aligned}
&\text{minimize } \nabla f_k^T s + \frac{1}{2}s^T H_k s \\
&\text{subject to } c_k + A_k^T s = 0 \\
&\quad x_k + s \geq 0
\end{aligned}
\]

2. Set \(x_{k+1} = x_k + s\)
Modern Interior Point Methods (IPM)

General NLP

\[
\begin{align*}
\text{minimize} & \quad f(x) \quad \text{subject to} \quad c(x) = 0 \quad \& \quad x \geq 0
\end{align*}
\]

Perturbed \( \mu > 0 \) optimality conditions \((x, z > 0)\)

\[
F_\mu(x, y, z) = \left\{ \begin{array}{l}
\nabla f(x) - \nabla c(x)^T y - z \\
\quad c(x) \\
Xz - \mu e
\end{array} \right\} = 0
\]

- Primal-dual formulation, where \( X = \text{diag}(x) \)
- Central path \( \{x(\mu), y(\mu), z(\mu) : \mu > 0\} \)
- Apply Newton’s method for sequence \( \mu \downarrow 0 \)
Modern Interior Point Methods (IPM)

Newton’s method applied to primal-dual system ...

\[
\begin{bmatrix}
\nabla^2 \mathcal{L}_k & -A_k & -I \\
A_k^T & 0 & 0 \\
Z_k & 0 & X_k
\end{bmatrix}
\begin{pmatrix}
\Delta x \\
\Delta y \\
\Delta z
\end{pmatrix}
= -F_\mu(x_k, y_k, z_k)
\]

where \( A_k = \nabla c(x_k)^T \), \( X_k \) diagonal matrix of \( x_k \).

Polynomial run-time guarantee for convex problems
Classical Interior Point Methods (IPM)

\[ \text{minimize } f(x) \quad \text{subject to } c(x) = 0 \quad \& \quad x \geq 0 \]

Related to classical barrier methods [Fiacco & McCormick]

\[ \begin{cases} 
\text{minimize } f(x) - \mu \sum \log(x_i) \\
\text{subject to } c(x) = 0 
\end{cases} \]

\[ \mu = 0.1 \quad \mu = 0.001 \]

\[ \text{minimize } x_1^2 + x_2^2 \quad \text{subject to } x_1 + x_2^2 \geq 1 \]
Classical Interior Point Methods (IPM)

\[
\minimize_x f(x) \quad \text{subject to } c(x) = 0 \ \& \ x \geq 0
\]

Relationship to barrier methods

\[
\begin{align*}
\minimize_x & \quad f(x) - \mu \sum \log(x_i) \\
\text{subject to } & \quad c(x) = 0
\end{align*}
\]

First order conditions

\[
\nabla f(x) - \mu X^{-1}e - A(x)y = 0 \\
c(x) = 0
\]

... apply Newton’s method ...
Classical Interior Point Methods (IPM)

Newton’s method for barrier problem from $x_k$ ...

\[
\begin{bmatrix}

\nabla^2 L_k + \mu X_k^{-2} & -A_k \\
A_k^T & 0
\end{bmatrix}
\begin{bmatrix}
\Delta x \\
\Delta y
\end{bmatrix} = ...
\]

Introduce $Z(x_k) := \mu X_k^{-1}$ ... or ...

\[
Z(x_k)X_k = \mu e
\]

\[
\begin{bmatrix}

\nabla^2 L_k + Z(x_k)X_k^{-1} & -A_k \\
A_k & 0
\end{bmatrix}
\begin{bmatrix}
\Delta x \\
\Delta y
\end{bmatrix} = ...
\]

... compare to primal-dual system ...
Classical Interior Point Methods (IPM)

Recall: Newton’s method applied to primal-dual system ...

\[
\begin{bmatrix}
\nabla^2 L_k & -A_k & -I \\
A_k^T & 0 & 0 \\
Z_k & 0 & X_k
\end{bmatrix}
\begin{pmatrix}
\Delta x \\
\Delta y \\
\Delta z
\end{pmatrix}
= - F_\mu(x_k, y_k, z_k)
\]

Eliminate \( \Delta z = -X^{-1}Z\Delta x - Ze - \mu X^{-1}e \)

\[
\begin{bmatrix}
\nabla^2 L_k + Z_kX_k^{-1} & -A_k \\
A_k & 0
\end{bmatrix}
\begin{pmatrix}
\Delta x \\
\Delta y
\end{pmatrix}
= ...
\]
Interior Point Methods (IPM)

Primal-dual system ...

\[
\begin{bmatrix}
\nabla^2 \mathcal{L}_k + Z_k X_k^{-1} & -A_k \\
A_k & 0
\end{bmatrix}
\begin{pmatrix}
\Delta x \\
\Delta y
\end{pmatrix} = ... 
\]

... compare to barrier system ...

\[
\begin{bmatrix}
\nabla^2 \mathcal{L}_k + Z(x_k) X_k^{-1} & -A_k \\
A_k & 0
\end{bmatrix}
\begin{pmatrix}
\Delta x \\
\Delta y
\end{pmatrix} = ... 
\]

- $Z_k$ is free, not $Z(x_k) = \mu X_k^{-1}$ (primal multiplier)
- avoid difficulties with barrier ill-conditioning
Convergence from Remote Starting Points

\[ \min_{x} f(x) \quad \text{subject to} \quad c(x) = 0 \quad \& \quad x \geq 0 \]

- Newton’s method converges \textit{quadratically} near a solution
- Newton’s method may \textit{diverge} if started far from solution
- How can we safeguard against this failure?

... motivates \textit{penalty} or \textit{merit functions} that

1. monitor progress towards a solution
2. combine objective \( f(x) \) and constraint violation \( \|c(x)\| \)
Penalty Functions (i)

Augmented Lagrangian Methods

\[ \min_{x} L(x, y_k, \rho_k) = f(x) - y_k^T c(x) + \frac{1}{2} \rho_k \| c(x) \|^2 \]

As \( y_k \to y_* \):
- \( x_k \to x_* \) for \( \rho_k > \bar{\rho} \)
- No ill-conditioning, improves convergence rate

- update \( \rho_k \) based on reduction in \( \| c(x) \|^2 \)
- approx. minimize \( L(x, y_k, \rho_k) \)
- first-order multiplier update: \( y_{k+1} = y_k - \rho_k c(x_k) \)
  \( \Rightarrow \) dual iteration
Penalty Functions (ii)

Exact Penalty Function

\[ \min_{x} \Phi(x, \pi) = f(x) + \pi \|c(x)\| \]

- combine constraints and objective
- equivalence of optimality \( \Rightarrow \) exact for \( \pi > \|y^*\|_D \)
  ... now apply unconstrained techniques
- \( \Phi \) nonsmooth, but equivalent to smooth problem (exercise)

... how do we enforce descent in merit functions???
Line Search Methods

SQP/IPM compute $s$ descend direction: $s^T \nabla \Phi < 0$

Backtracking-Armijo line search

Given $\alpha^0 = 1$, $\beta = 0.1$, set $l = 0$

**REPEAT**

1. $\alpha^{l+1} = \alpha^l/2$ & evaluate $\Phi(x + \alpha^{l+1}s)$
2. $l = l + 1$

**UNTIL** $\Phi(x + \alpha^l s) \leq f(x) + \alpha^l \beta s^T \nabla \Phi$

Converges to stationary point, or unbounded, or zero descend
Line Search Methods

\[ f(x_k) \]

\[ f(x_k + ts) \]

acceptable
Trust Region Methods

Globalize SQP (IPM) by adding trust region, $\Delta^k > 0$

$$\begin{align*}
\text{minimize} & \quad \nabla f_k^T s + \frac{1}{2} s^T H_k s \\
\text{subject to} & \quad c_k + A_k^T s = 0, \quad x_k + s \geq 0, \quad \|s\| \leq \Delta^k
\end{align*}$$
Trust Region Methods

Globalize SQP (IPM) by adding trust region, $\Delta^k > 0$

$$\begin{aligned}
\text{minimize} & \quad \nabla f_k^T s + \frac{1}{2} s^T H_k s \\
\text{subject to} & \quad c_k + A_k^T s = 0, \quad x_k + s \geq 0, \quad \|s\| \leq \Delta^k
\end{aligned}$$

REPEAT

1. Solve QP approximation about $x_k$
2. Compute actual/predicted reduction, $r_k$
3. IF $r_k \geq 0.75$ THEN $x_{k+1} = x_k + s$
   ELSEIF $r_k \geq 0.25$ THEN $x_{k+1} = x_k + s$
   ELSE $x_{k+1} = x_k$ & decrease $\Delta$

UNTIL convergence
Filter Methods for NLP

Penalty function can be inefficient

- Penalty parameter *not known a priori*
- Large penalty parameter $\Rightarrow$ slow convergence
Filter Methods for NLP

Penalty function can be inefficient

- Penalty parameter not known a priori
- Large penalty parameter $\Rightarrow$ slow convergence

Two competing aims in optimization:

1. Minimize $f(x)$
2. Minimize $h(x) := \|c(x)\|$ ... more important

$\Rightarrow$ concept from multi-objective optimization:

$(h_{k+1}, f_{k+1})$ dominates $(h_l, f_l)$ iff $h_{k+1} \leq h_l$ & $f_{k+1} \leq f_l$
Filter Methods for NLP

Filter $\mathcal{F}$: list of non-dominated pairs $(h_l, f_l)$

- new $x_{k+1}$ acceptable to filter $\mathcal{F}$, iff
  1. $h_{k+1} \leq h_l \quad \forall l \in \mathcal{F}$, or
  2. $f_{k+1} \leq f_l \quad \forall l \in \mathcal{F}$
- remove redundant entries
- reject new $x_{k+1}$,
  if $h_{k+1} > h_l$ & $f_{k+1} > f_l$
  ... reduce trust region radius $\Delta = \Delta / 2$

$\Rightarrow$ often accept new $x_{k+1}$, even if penalty function increases
Sequential Quadratic Programming

- **filterSQP**
  - trust-region SQP; robust QP solver
  - filter to promote global convergence
- **SNOPT**
  - line-search SQP; null-space CG option
  - $\ell_1$ exact penalty function
- **SLIQUE (part of KNITRO)**
  - SLP-EQP (“SQP” for larger problems)
  - trust-region with $\ell_1$ penalty

Other Methods: CONOPT generalized reduced gradient method
Interior Point Methods

- **IPOPT** (free: part of COIN-OR)
  - line-search filter algorithm
  - 2nd order convergence analysis for filter
- **KNITRO**
  - trust-region Newton to solve barrier problem
  - $\ell_1$ penalty barrier function
  - Newton system: direct solves or null-space CG
- **LOQO**
  - line-search method
  - Cholesky factorization; no convergence analysis

Other solvers: **MOSEK** (unsuitable or nonconvex problem)
Augmented Lagrangian Methods

- **LANCELOT**
  - minimize augmented Lagrangian subject to bounds
  - trust-region to force convergence
  - iterative (CG) solves

- **MINOS**
  - minimize augmented Lagrangian subject to linear constraints
  - line-search; recent convergence analysis
  - direct factorization of linear constraints

- **PENNON**
  - suitable for semi-definite optimization
  - alternative penalty terms
Automatic Differentiation

How do I get the derivatives $\nabla c(x)$, $\nabla^2 c(x)$ etc?

- hand-coded derivatives are error prone
- finite differences $\frac{\partial c_i(x)}{\partial x_j} \approx \frac{c_i(x+\delta e_j)-c_i(x)}{\delta}$ can be dangerous

where $e_j = (0, \ldots, 0, 1, 0, \ldots, 0)$ is $j^{th}$ unit vector

Automatic Differentiation

- chain rule techniques to differentiate program
- recursive application $\Rightarrow$ “exact” derivatives
- suitable for huge problems, see www.autodiff.org

... already done for you in AMPL/GAMS etc.
1. floating point (IEEE) exceptions
2. unbounded problems
   2.1 unbounded objective
   2.2 unbounded multipliers
3. (locally) inconsistent problems
4. suboptimal solutions

... identify problem & suggest remedies
Floating Point (IEEE) Exceptions

Bad example: minimize barrier function

```plaintext
param mu default 1;
var x{1..2} >= -10, <= 10;
var s;
minimize barrier: x[1]^2 + x[2]^2 - mu*log(s);
subject to
```

... results in error message like

```
Cannot evaluate objective at start
```

... change initialization of \( s \):

```plaintext
var s := 1; ... difficult, if IEEE during solve ...
```
Unbounded Objective

Penalized MPEC $\pi = 1$:

$$\begin{align*}
\text{minimize} & \quad x_1^2 + x_2^2 - 4x_1x_2 + \pi x_1x_2 \\
\text{subject to} & \quad x_1, x_2 \geq 0
\end{align*}$$

... unbounded below for all $\pi < 2$
Locally Inconsistent Problems

NLP may have no feasible point

feasible set: intersection of circles
Locally Inconsistent Problems

NLP may have no feasible point

var x{1..2} >= -1;
minimize objf: -1000*x[2];
subject to
  con1: (x[1]+2)^2 + x[2]^2 <= 1;
  con2: (x[1]-2)^2 + x[2]^2 <= 1;

• not all solvers recognize this ...
• finding feasible point ⇔ global optimization
Locally Inconsistent Problems

**LOQO**

<table>
<thead>
<tr>
<th>Iter</th>
<th>Primal</th>
<th>Dual</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Obj Value</td>
<td>Infeas</td>
</tr>
<tr>
<td>1</td>
<td>-1.000000e+03</td>
<td>4.2e+00</td>
</tr>
<tr>
<td>500</td>
<td>2.312535e-04</td>
<td>7.9e-01</td>
</tr>
</tbody>
</table>

LOQO 6.06: iteration limit

... fails to converge ... not useful for user

dual unbounded $\rightarrow \infty \Rightarrow$ primal infeasible
Locally Inconsistent Problems

FILTER

| iter | rho  | ||d||    | f / hJ   | ||c||/hJt |
|------|------|---------|----------|----------|
| 0:0  | 10.0000 | 0.00000 | -1000.0000 | 16.000000 |
| 1:1  | 10.0000 | 2.00000 | -1000.0000 | 8.0000000 |
| [...]|
| 8:2  | 2.00000 | 0.320001E-02 | 7.9999693 | 0.10240052E-04 |
| 9:2  | 2.00000 | 0.512000E-05 | 8.0000000 | 0.26214586E-10 |

filterSQP: Nonlinear constraints locally infeasible

... fast convergence to minimum infeasibility
... identify “blocking” constraints ... modify model/data
Locally Inconsistent Problems

Remedies for locally infeasible problems:

1. check your model: print constraints & residuals, e.g. solve;
   display _conname, _con.lb, _con.body, _con.ub;
   display _varname, _var.lb, _var, _var.ub;
   ... look at violated and active constraints

2. try different nonlinear solvers (easy with AMPL)

3. build-up model from few constraints at a time

4. try different starting points ... global optimization
Suboptimal Solution & Multi-start

Problems can have many local minimizers

```
param pi := 3.1416;
param n integer, >= 0, default 2;
set N := 1..n;
var x{N} >= 0, <= 32*pi, := 1;
minimize objf:
- sum{i in N} x[i]*sin(sqrt(x[i]));
```

default start point converges to local minimizer
Suboptimal Solution & Multi-start

```plaintext
param nD := 5;       # discretization
set D := 1..nD;
param hD := 32*pi/(nD-1);
param optval{D,D};
model schwefel.mod;  # load model

for {i in D}{
    let x[1] := (i-1)*hD;
    for {j in D}{
        let x[2] := (j-1)*hD;
        solve;
        let optval[i,j] := objf;
    }
}; # end for
}; # end for
```
Suboptimal Solution & Multi-start

\[
\text{display optval;}
\text{optval [*,*]}
\begin{array}{cccccc}
\text{:} & 1 & 2 & 3 & 4 & 5 & := \\
1 & 0 & 24.003 & -36.29 & -50.927 & 56.909 \\
3 & -36.29 & -67.5803 & -127.27 & -127.27 & -127.27 \\
4 & -50.927 & -67.5803 & -127.27 & -127.27 & -127.27 \\
5 & 56.909 & -67.5803 & -127.27 & -127.27 & -127.27 \\
\end{array}
\]

... there exist better multi-start procedures
Optimization with Integer Variables

- modeling discrete choices $\Rightarrow$ 0–1 variables
- modeling integer decisions $\Rightarrow$ integer variables
  e.g. number of different stocks in portfolio (8-10)
  not number of beers sold at Goose Island (millions)

$\Rightarrow$ Mixed Integer Nonlinear Program (MINLP)

MINLP solvers:
- branch (separate $z_i = 0$ and $z_i = 1$) and cut
- solve millions of NLP relaxations: MINLPBB, SBB
- outer approximation: iterate MILP and NLP solvers
  BONMIN soon on COIN-OR
Portfolio Management

- **\( N \)**: Universe of asset to purchase
- **\( x_i \)**: Amount of asset \( i \) to hold
- **\( B \)**: Budget

\[
\min_{x \in \mathbb{R}^{\lvert N \rvert}} \left\{ u(x) \mid \sum_{i \in N} x_i = B \right\}
\]

- **Markowitz**: \( u(x) \stackrel{\text{def}}{=} -\alpha^T x + \lambda x^T Q x \)
  - **\( \alpha \)**: Expected returns
  - **\( Q \)**: Variance-covariance matrix of expected returns
  - **\( \lambda \)**: Risk aversion parameter
More Realistic Models

- $b \in \mathbb{R}^{|N|}$ of “benchmark” holdings
- Benchmark Tracking: $u(x) \overset{\text{def}}{=} (x - b)^T Q (x - b)$
  - Constraint on $\mathbb{E}[\text{Return}]: \alpha^T x \geq r$
- Limit Names: $|i \in N : x_i > 0| \leq K$
  - Use binary indicator variables to model the implication $x_i > 0 \Rightarrow y_i = 1$
  - Implication modeled with variable upper bounds:
    \[ x_i \leq B y_i \quad \forall i \in N \]
- $\sum_{i \in N} y_i \leq K$
Even More Models

- **Min Holdings**: \((x_i = 0) \lor (x_i \geq m)\)
  - Model implication: \(x_i > 0 \Rightarrow x_i \geq m\)
  - \(x_i > 0 \Rightarrow y_i = 1 \Rightarrow x_i \geq m\)
  - \(x_i \leq By_i, x_i \geq my_i \forall i \in N\)

- **Round Lots**: \(x_i \in \{kL_i, k = 1, 2, \ldots\}\)
  - \(x_i - z_iL_i = 0, z_i \in \mathbb{Z}_+ \forall i \in N\)

- **Vector** \(h\) of initial holdings
- **Transactions**: \(t_i = |x_i - h_i|\)
- **Turnover**: \(\sum_{i \in N} t_i \leq \Delta\)
- **Transaction Costs**: \(\sum_{i \in N} c_i t_i\) in objective
- **Market Impact**: \(\sum_{i \in N} \gamma_i t_i^2\) in objective
Global Optimization

I need to find the GLOBAL minimum!

- use any NLP solver (often work well!)
- use the multi-start trick from previous slides
- global optimization based on branch-and-reduce: BARON
  - constructs global underestimators
  - refines region by branching
  - tightens bounds by solving LPs
  - solve problems with 100s of variables
- “voodoo” solvers: genetic algorithm & simulated annealing
  no convergence theory ... usually worse than deterministic
Derivative-Free Optimization

My model does not have derivatives!

- Change your model ... good models have derivatives!
- pattern-search methods for \( \min f(x) \)
  - evaluate \( f(x) \) at stencil \( x_k + \Delta M \)
  - move to new best point
  - extend to NLP; some convergence theory
- matlab: NOMADm.m; parallel APPSPACK
- solvers based on building quadratic models
- “voodoo” solvers: genetic algorithm & simulated annealing
  no convergence theory ... usually worse than deterministic
COIN-OR

http://www.coin-or.org

- **CO**mputational **IN**frastructure for **O**perations **R**esearch
- A **l**ibrary of (interoperable) software tools for optimization
- A **d**evelopment **p**latform for open source projects in the OR community
- Possibly Relevant Modules:
  - OSI: **O**pen **S**olver **I**nterface
  - CGL: **C**ut **G**eneration **L**ibrary
  - CLP: **C**oin **L**inear **P**rogramming **T**oolkit
  - CBC: **C**oin **B**ranch and **C**ut
  - IPOPT: **I**nterior **P**oint **OPT**imizer for NLP
  - NLPAPI: **N**on**L**inear **P**rogramming **API**
Part IV

Complementarity Problems in AMPL
Definition

- Non-cooperative game played by \( n \) individuals
  - Each player selects a strategy to optimize their objective
  - Strategies for the other players are fixed
- Equilibrium reached when no improvement is possible
- Characterization of two player equilibrium \((x^*, y^*)\)

\[
\begin{align*}
x^* &\in \left\{ \begin{array}{ll}
\arg\min_{x \geq 0} & f_1(x, y^*) \\
\text{subject to} & c_1(x) \leq 0
\end{array} \right. \\
y^* &\in \left\{ \begin{array}{ll}
\arg\min_{y \geq 0} & f_2(x^*, y) \\
\text{subject to} & c_2(y) \leq 0
\end{array} \right.
\end{align*}
\]

- Many applications in economics
  - Bi-matrix games
  - Cournot duopoly models
  - General equilibrium models
  - Arrow-Debreu models
Complementarity Formulation

• Assume each optimization problem is convex
  • \( f_1(\cdot, y) \) is convex for each \( y \)
  • \( f_2(x, \cdot) \) is convex for each \( x \)
  • \( c_1(\cdot) \) and \( c_2(\cdot) \) satisfy constraint qualification
• Then the first-order conditions are necessary and sufficient

\[
\begin{align*}
\min_{x \geq 0} & \quad f_1(x, y^*) \\
\text{subject to} & \quad c_1(x) \leq 0
\end{align*}
\iff
\begin{align*}
0 \leq x \perp & \nabla_x f_1(x, y^*) + \lambda_1^T \nabla_x c_1(x) \geq 0 \\
0 \leq \lambda_1 \perp & -c_1(x) \geq 0
\end{align*}
\]

\[
\begin{align*}
\min_{y \geq 0} & \quad f_2(x^*, y) \\
\text{subject to} & \quad c_2(y) \leq 0
\end{align*}
\iff
\begin{align*}
0 \leq y \perp & \nabla_y f_2(x^*, y) + \lambda_2^T \nabla_y c_2(y) \geq 0 \\
0 \leq \lambda_2 \perp & -c_2(y) \geq 0
\end{align*}
\]
Formulation

- Firm $f \in \mathcal{F}$ chooses output $x_f$ to maximize profit
  - $u$ is the utility function
  - $u = \left( 1 + \sum_{f \in \mathcal{F}} x_f^\alpha \right)^{\frac{n}{\alpha}}$
  - $\alpha$ and $\eta$ are parameters
  - $c_f$ is the unit cost for each firm
- In particular, for each firm $f \in \mathcal{F}$
  - $x^*_f \in \arg \max_{x_f \geq 0} \left( \frac{\partial u}{\partial x_f} - c_f \right) x_f$
- First-order optimality conditions
  - $0 \leq x_f \perp c_f - \frac{\partial u}{\partial x_f} - x_f \frac{\partial^2 u}{\partial x^2_f} \geq 0$
Model: oligopoly.mod

```plaintext
set FIRMS;  # Firms in problem

param c {FIRMS};  # Unit cost
param alpha > 0;  # Constants
param eta > 0;

var x {FIRMS} default 0.1;  # Output (no bounds!)

var s = 1 + sum {f in FIRMS} x[f]^alpha;  # Summation term
var u = s^(eta/alpha);  # Utility
var du {f in FIRMS} = eta * s^(eta/alpha - 1) * x[f]^(alpha - 1);  # Marginal price
var dudu {f in FIRMS} = eta * (eta - alpha) * s^(eta/alpha - 2) * x[f]^(2 * alpha - 2) + eta * (alpha - 1) * s^(eta/alpha - 1) * x[f]^(alpha - 2);  # Derivative

compl {f in FIRMS}:
  0 <= x[f] complements c[f] - du[f] - x[f] * dudu[f] >= 0;
```

Leyffer, Moré, and Munson
Computational Optimization
Data: oligopoly.dat

param: FIRMS : c :=
1 0.07
2 0.08
3 0.09;

param alpha := 0.999;
param eta := 0.2;
Commands: oligopoly.cmd

# Load model and data
model oligopoly.mod;
data oligopoly.dat;

# Specify solver and options
option presolve 0;
option solver "pathampl";

# Solve complementarity problem
solve;

# Output the results
printf {f in FIRMS} "Output for firm %2d: % 5.4e\n", f, x[f] > oligcomp.out;
Results: oligopoly.out

Output for firm 1: 8.3735e-01
Output for firm 2: 5.0720e-01
Output for firm 3: 1.7921e-01
Model Formulation

- Economy with $n$ agents and $m$ commodities
  - $e \in \mathbb{R}^{n \times m}$ are the endowments
  - $\alpha \in \mathbb{R}^{n \times m}$ and $\beta \in \mathbb{R}^{n \times m}$ are the utility parameters
  - $p \in \mathbb{R}^m$ are the commodity prices
- Agent $i$ maximizes utility with budget constraint
  \[
  \max_{x_i,* \geq 0} \sum_{k=1}^m \frac{\alpha_{i,k} (1 + x_{i,k})^{1-\beta_{i,k}}}{1 - \beta_{i,k}}
  \]
  subject to
  \[
  \sum_{k=1}^m p_k (x_{i,k} - e_{i,k}) \leq 0
  \]
- Market $k$ sets price for the commodity
  \[
  0 \leq p_k \perp \sum_{i=1}^n (e_{i,k} - x_{i,k}) \geq 0
  \]
Model: cge.mod

set AGENTS;                  # Agents
set COMMODITIES;             # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param alpha {AGENTS, COMMODITIES} > 0;         # Utility parameters
param beta {AGENTS, COMMODITIES} > 0;

var x {AGENTS, COMMODITIES};                    # Consumption (no bounds!)
var l {AGENTS};                                # Multipliers (no bounds!)
var p {COMMODITIES};                           # Prices (no bounds!)

var du {i in AGENTS, k in COMMODITIES} =        # Marginal prices
  alpha[i,k] / (1 + x[i,k])^beta[i,k];

subject to
  optimality {i in AGENTS, k in COMMODITIES}:
    0 <= x[i,k] complements -du[i,k] + p[k] * l[i] >= 0;

  budget {i in AGENTS}:
    0 <= l[i] complements sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

  market {k in COMMODITIES}:
    0 <= p[k] complements sum {i in AGENTS} (e[i,k] - x[i,k]) >= 0;
Data: cge.dat

set AGENTS := Jorge, Sven, Todd;
set COMMODITIES := Books, Cars, Food, Pens;

param alpha : Books Cars Food Pens :=
  Jorge  1  1  1  1
  Sven   1  2  3  4
  Todd   2  1  1  5;

param beta (tr): Jorge Sven Todd :=
  Books  1.5  2  0.6
  Cars   1.6  3  0.7
  Food   1.7  2  2.0
  Pens   1.8  2  2.5;
# Load model and data
model cge.mod;
data cge.dat;

# Specify solver and options
option presolve 0;
option solver "pathampl";

# Solve the instance
solve;

# Output results
printf {i in AGENTS, k in COMMODITIES} "%s %5s: % 5.4e\n", i, k, x[i,k] > cge.out;
printf "\n" > cge.out;
printf {k in COMMODITIES} "%s: % 5.4e\n", k, p[k] > cge.out;
## Results: `cge.out`

<table>
<thead>
<tr>
<th>Item</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>Jorge Books</td>
<td>8.9825e-01</td>
</tr>
<tr>
<td>Jorge Cars</td>
<td>1.4651e+00</td>
</tr>
<tr>
<td>Jorge Food</td>
<td>1.2021e+00</td>
</tr>
<tr>
<td>Jorge Pens</td>
<td>6.8392e-01</td>
</tr>
<tr>
<td>Sven Books</td>
<td>2.5392e-01</td>
</tr>
<tr>
<td>Sven Cars</td>
<td>7.2054e-01</td>
</tr>
<tr>
<td>Sven Food</td>
<td>1.6271e+00</td>
</tr>
<tr>
<td>Sven Pens</td>
<td>1.4787e+00</td>
</tr>
<tr>
<td>Todd Books</td>
<td>1.8478e+00</td>
</tr>
<tr>
<td>Todd Cars</td>
<td>8.1431e-01</td>
</tr>
<tr>
<td>Todd Food</td>
<td>1.7081e-01</td>
</tr>
<tr>
<td>Todd Pens</td>
<td>8.3738e-01</td>
</tr>
<tr>
<td>Books</td>
<td>1.0825e+01</td>
</tr>
<tr>
<td>Cars</td>
<td>6.6835e+00</td>
</tr>
<tr>
<td>Food</td>
<td>7.3983e+00</td>
</tr>
<tr>
<td>Pens</td>
<td>1.1081e+01</td>
</tr>
</tbody>
</table>
# Load model and data
model cge.mod;
data cge.dat;

# Specify solver and options
option presolve 0;
option solver "pathampl";

# Solve the instance
drop market['Books'];
fix p['Books'] := 1;
solve;

# Output results
printf {i in AGENTS, k in COMMODITIES} "%5s %5s: % 5.4e\n", i, k, x[i,k] > cgenum.out;
printf "\n" > cgenum.out;
printf {k in COMMODITIES} "%5s: % 5.4e\n", k, p[k] > cgenum.out;
Results: cgenum.out

Jorge Books: \(8.9825 \times 10^{-1}\)
Jorge Cars: \(1.4651 \times 10^{0}\)
Jorge Food: \(1.2021 \times 10^{0}\)
Jorge Pens: \(6.8392 \times 10^{-1}\)
Sven Books: \(2.5392 \times 10^{-1}\)
Sven Cars: \(7.2054 \times 10^{-1}\)
Sven Food: \(1.6271 \times 10^{0}\)
Sven Pens: \(1.4787 \times 10^{0}\)
Todd Books: \(1.8478 \times 10^{0}\)
Todd Cars: \(8.1431 \times 10^{-1}\)
Todd Food: \(1.7081 \times 10^{-1}\)
Todd Pens: \(8.3738 \times 10^{-1}\)

Books: \(1.0000 \times 10^{0}\)
Cars: \(6.1742 \times 10^{-1}\)
Food: \(6.8345 \times 10^{-1}\)
Pens: \(1.0237 \times 10^{0}\)
Formulation

- Players select strategies to minimize loss
  - $p \in \mathbb{R}^n$ is the probability player 1 chooses each strategy
  - $q \in \mathbb{R}^m$ is the probability player 2 chooses each strategy
  - $A \in \mathbb{R}^{n \times m}$ is the loss matrix for player 1
  - $B \in \mathbb{R}^{n \times m}$ is the loss matrix for player 2

- Optimization problem for player 1
  \[
  \min_{0 \leq p \leq 1} \quad p^T A q \\
  \text{subject to} \quad e^T p = 1
  \]

- Optimization problem for player 2
  \[
  \min_{0 \leq q \leq 1} \quad p^T B q \\
  \text{subject to} \quad e^T q = 1
  \]

- Complementarity problem
  \[
  0 \leq p \leq 1 \perp \quad A q - \lambda_1 \\
  \text{subject to} \quad B^T p - \lambda_2
  \]
Model: bimatrix1.mod

param n > 0, integer;  # Strategies for player 1
param m > 0, integer;  # Strategies for player 2

param A{1..n, 1..m};  # Loss matrix for player 1
param B{1..n, 1..m};  # Loss matrix for player 2

var p{1..n};  # Probability player 1 selects strategy i
var q{1..m};  # Probability player 2 selects strategy j
var lambda1;  # Multiplier for constraint
var lambda2;  # Multiplier for constraint

subject to
    opt1 {i in 1..n}:  # Optimality conditions for player 1
        0 <= p[i] <= 1 complements sum{j in 1..m} A[i,j] * q[j] - lambda1;

    opt2 {j in 1..m}:  # Optimality conditions for player 2
        0 <= q[j] <= 1 complements sum{i in 1..n} B[i,j] * p[i] - lambda2;

    con1:  
            lambda1 complements sum{i in 1..n} p[i] = 1;

    con2:  
            lambda2 complements sum{j in 1..m} q[j] = 1;
Model: bimatrix2.mod

param n > 0, integer;  # Strategies for player 1
param m > 0, integer;  # Strategies for player 2

param A{1..n, 1..m};  # Loss matrix for player 1
param B{1..n, 1..m};  # Loss matrix for player 2

var p{1..n};  # Probability player 1 selects strategy i
var q{1..m};  # Probability player 2 selects strategy j
var lambda1;  # Multiplier for constraint
var lambda2;  # Multiplier for constraint

subject to
  opt1 {i in 1..n}:  # Optimality conditions for player 1
    0 <= p[i] complements sum{j in 1..m} A[i,j] * q[j] - lambda1 >= 0;
  opt2 {j in 1..m}:  # Optimality conditions for player 2
    0 <= q[j] complements sum{i in 1..n} B[i,j] * p[i] - lambda2 >= 0;
  con1:
    0 <= lambda1 complements sum{i in 1..n} p[i] >= 1;
  con2:
    0 <= lambda2 complements sum{j in 1..m} q[j] >= 1;
Model: bimatrix3.mod

param n > 0, integer; # Strategies for player 1
param m > 0, integer; # Strategies for player 2

param A{1..n, 1..m}; # Loss matrix for player 1
param B{1..n, 1..m}; # Loss matrix for player 2

var p{1..n}; # Probability player 1 selects strategy i
var q{1..m}; # Probability player 2 selects strategy j

subject to
  opt1 {i in 1..n}: # Optimality conditions for player 1
    0 <= p[i] complements sum{j in 1..m} A[i,j] * q[j] >= 1;

  opt2 {j in 1..m}: # Optimality conditions for player 2
    0 <= q[j] complements sum{i in 1..n} B[i,j] * p[i] >= 1;
Pitfalls

- Nonsquare systems
  - Side variables
  - Side constraints
- Orientation of equations
  - Skew symmetry preferred
  - Proximal point perturbation
- AMPL presolve
  - option presolve 0;
Definition

- Leader-follower game
  - Dominant player (leader) selects a strategy $y^*$
  - Then followers respond by playing a Nash game

\[
x_i^* \in \left\{ \arg \min_{x_i \geq 0} f_i(x, y) \right. \\
\left. \text{subject to } c_i(x_i) \leq 0 \right\}
\]

- Leader solves optimization problem with equilibrium constraints

\[
\min_{y \geq 0, x, \lambda} g(x, y) \\
\text{subject to } h(y) \leq 0 \\
0 \leq x_i \perp \nabla x_i f_i(x, y) + \lambda_i^T \nabla x_i c_i(x_i) \geq 0 \\
0 \leq \lambda_i \perp -c_i(x_i) \geq 0
\]

- Many applications in economics
  - Optimal taxation
  - Tolleing problems
Nonlinear Programming Formulation

\[
\begin{align*}
\min_{x, y, \lambda, s, t \geq 0} & \quad g(x, y) \\
\text{subject to} & \quad h(y) \leq 0 \\
& \quad s_i = \nabla_x f_i(x, y) + \lambda_i^T \nabla_x c_i(x_i) \\
& \quad t_i = -c_i(x_i) \\
& \quad \sum_i (s_i^T x_i + \lambda_i t_i) \leq 0
\end{align*}
\]

- Constraint qualification fails
  - Lagrange multiplier set unbounded
  - Constraint gradients linearly dependent
  - Central path does not exist
- Able to prove convergence results for some methods
- Reformulation very successful and versatile in practice
Penalization Approach

\[
\begin{align*}
\min_{x,y,\lambda,s,t \geq 0} & \quad g(x,y) + \pi \sum_i (s_i^T x_i + \lambda_i t_i) \\
\text{subject to} & \quad h(y) \leq 0 \\
& \quad s_i = \nabla_{x_i} f_i(x,y) + \lambda_i^T \nabla_{x_i} c_i(x_i) \\
& \quad t_i = -c_i(x_i)
\end{align*}
\]

- Optimization problem satisfies constraint qualification
- Need to increase $\pi$
Relaxation Approach

\[
\begin{align*}
\min_{x,y,\lambda,s,t \geq 0} & \quad g(x, y) \\
\text{subject to} & \quad h(y) \leq 0 \\
& \quad s_i = \nabla x_i f_i(x, y) + \lambda_i^T \nabla x_i c_i(x_i) \\
& \quad t_i = -c_i(x_i) \\
& \quad \sum_i (s_i^T x_i + \lambda_i t_i) \leq \tau
\end{align*}
\]

- Need to decrease \( \tau \)
Model Formulation

- Economy with $n$ agents and $m$ commodities
  - $e \in \mathbb{R}^{n \times m}$ are the endowments
  - $\alpha \in \mathbb{R}^{n \times m}$ and $\beta \in \mathbb{R}^{n \times m}$ are the utility parameters
  - $p \in \mathbb{R}^m$ are the commodity prices
- Agent $i$ maximizes utility with budget constraint

$$\max_{x_{i,*} \geq 0} \sum_{k=1}^{m} \frac{\alpha_{i,k}(1 + x_{i,k})^{1-\beta_{i,k}}}{1 - \beta_{i,k}}$$

subject to

$$\sum_{k=1}^{m} p_k (x_{i,k} - e_{i,k}) \leq 0$$

- Market $k$ sets price for the commodity

$$0 \leq p_k \perp \sum_{i=1}^{n} (e_{i,k} - x_{i,k}) \geq 0$$
Model: cgempec.mod

```plaintext
set LEADER; # Leader
set FOLLOWERS; # Followers
set AGENTS := LEADER union FOLLOWERS; # All the agents
check: (card(LEADER) == 1 && card(LEADER inter FOLLOWERS) == 0);

set COMMODITIES; # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param alpha {AGENTS, COMMODITIES} > 0; # Utility parameters
param beta {AGENTS, COMMODITIES} > 0;

var x {AGENTS, COMMODITIES}; # Consumption (no bounds!)
var l {FOLLOWERS}; # Multipliers (no bounds!)
var p {COMMODITIES}; # Prices (no bounds!)

var u {i in AGENTS} = # Utility
    sum {k in COMMODITIES} alpha[i,k] * (1 + x[i,k])^(1 - beta[i,k]) / (1 - beta[i,k]);

var du {i in AGENTS, k in COMMODITIES} = # Marginal prices
    alpha[i,k] / (1 + x[i,k])^beta[i,k];
```
Model: cgempec.mod

maximize
   objective: sum {i in LEADER} u[i];

subject to
   leader_budget {i in LEADER}:
      sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

   optimality {i in FOLLOWERS, k in COMMODITIES}:
      0 <= x[i,k] complements -du[i,k] + p[k] * l[i] >= 0;

   budget {i in FOLLOWERS}:
      0 <= l[i] complements sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

   market {k in COMMODITIES}:
      0 <= p[k] complements sum {i in AGENTS} (e[i,k] - x[i,k]) >= 0;
Data: cgempec.dat

set LEADER := Jorge;
set FOLLOWERS := Sven, Todd;
set COMMODITIES := Books, Cars, Food, Pens;

param alpha : Books Cars Food Pens :=
    Jorge       1   1   1   1
    Sven        1   2   3   4
    Todd        2   1   1   5;

param beta (tr): Jorge Sven Todd :=
    Books      1.5  2   0.6
    Cars       1.6  3   0.7
    Food       1.7  2   2.0
    Pens       1.8  2   2.5;
Commands: cgempec.cmd

# Load model and data
model cgempec.mod;
data cgempec.dat;

# Specify solver and options
option presolve 0;
option solver "loqo";

# Solve the instance
drop market['Books'];
fix p['Books'] := 1;
solve;

# Output results
printf {i in AGENTS, k in COMMODITIES} "%5s %5s: % 5.4e\n", i, k, x[i,k] > cgempec.out;
printf "\n" > cgempec.out;
printf {k in COMMODITIES} "%5s: % 5.4e\n", k, p[k] > cgempec.out;
### Output: cgempec.out

<table>
<thead>
<tr>
<th></th>
<th>Stackleberg</th>
<th>Nash Game</th>
</tr>
</thead>
<tbody>
<tr>
<td>Jorge Cars</td>
<td>1.3666e+00</td>
<td>Jorge Cars: 1.4651e+00</td>
</tr>
<tr>
<td>Jorge Food</td>
<td>1.1508e+00</td>
<td>Jorge Food: 1.2021e+00</td>
</tr>
<tr>
<td>Jorge Pens</td>
<td>7.7259e-01</td>
<td>Jorge Pens: 6.8392e-01</td>
</tr>
<tr>
<td>Sven Books</td>
<td>2.5499e-01</td>
<td>Sven Books: 2.5392e-01</td>
</tr>
<tr>
<td>Sven Cars</td>
<td>7.4173e-01</td>
<td>Sven Cars: 7.2054e-01</td>
</tr>
<tr>
<td>Sven Food</td>
<td>1.6657e+00</td>
<td>Sven Food: 1.6271e+00</td>
</tr>
<tr>
<td>Sven Pens</td>
<td>1.4265e+00</td>
<td>Sven Pens: 1.4787e+00</td>
</tr>
<tr>
<td>Todd Books</td>
<td>1.8205e+00</td>
<td>Todd Books: 1.8478e+00</td>
</tr>
<tr>
<td>Todd Cars</td>
<td>8.9169e-01</td>
<td>Todd Cars: 8.1431e-01</td>
</tr>
<tr>
<td>Todd Food</td>
<td>1.8355e-01</td>
<td>Todd Food: 1.7081e-01</td>
</tr>
<tr>
<td>Todd Pens</td>
<td>8.0093e-01</td>
<td>Todd Pens: 8.3738e-01</td>
</tr>
<tr>
<td>Books</td>
<td>1.0000e+00</td>
<td>Books: 1.0000e+00</td>
</tr>
<tr>
<td>Cars</td>
<td>5.9617e-01</td>
<td>Cars: 6.1742e-01</td>
</tr>
<tr>
<td>Food</td>
<td>6.6496e-01</td>
<td>Food: 6.8345e-01</td>
</tr>
<tr>
<td>Pens</td>
<td>1.0700e+00</td>
<td>Pens: 1.0237e+00</td>
</tr>
</tbody>
</table>
Unbounded Multipliers

\[
\text{var } z\{1..2\} \geq 0;
\]
\[
\text{var } z3;
\]

\[
\text{minimize } \text{objf}: z[1] + z[2] - z3;
\]

\[
\text{subject to}
\]
\[
\text{lin1}: \quad -4 \times z[1] + z3 \leq 0;
\]
\[
\text{lin2}: \quad -4 \times z[2] + z3 \leq 0;
\]
\[
\text{compl}: \quad z[1] \times z[2] \leq 0;
\]
## LOQO Output

**LOQO 6.06: outlev=2**

<table>
<thead>
<tr>
<th>Iter</th>
<th>Primal</th>
<th>Dual</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Obj Value</td>
<td>Infeas</td>
</tr>
<tr>
<td>1</td>
<td>1.000000e+00</td>
<td>0.0e+00</td>
</tr>
<tr>
<td>2</td>
<td>6.902180e-01</td>
<td>2.2e-01</td>
</tr>
<tr>
<td>3</td>
<td>2.773222e-01</td>
<td>1.6e-01</td>
</tr>
<tr>
<td>292</td>
<td>-8.213292e-05</td>
<td>1.7e-09</td>
</tr>
<tr>
<td>293</td>
<td>-8.202525e-05</td>
<td>1.7e-09</td>
</tr>
<tr>
<td>294</td>
<td>nan</td>
<td>nan</td>
</tr>
<tr>
<td>500</td>
<td>nan</td>
<td>nan</td>
</tr>
<tr>
<td>iter</td>
<td>rho</td>
<td></td>
</tr>
<tr>
<td>------</td>
<td>-------</td>
<td>------</td>
</tr>
<tr>
<td>0:0</td>
<td>10.000</td>
<td>0.000</td>
</tr>
<tr>
<td>1:1</td>
<td>10.000</td>
<td>1.000</td>
</tr>
<tr>
<td>24:1</td>
<td>0.156</td>
<td>0.197E-05</td>
</tr>
<tr>
<td>25:1</td>
<td>0.156</td>
<td>0.983477E-06</td>
</tr>
</tbody>
</table>

Norm of KKT residual: 0.471404521
max(|\lambda_i| * ||a_i||): 2.06155281
Largest modulus multiplier: 2711469.25
Limitations

- Multipliers may not exist
- Solvers can have a hard time computing solutions
  - Try different algorithms
  - Compute feasible starting point
- Stationary points may have descent directions
  - Checking for descent is an exponential problem
  - Strong stationary points found in certain cases
- Many stationary points – global optimization
- Formulation of follower problem
  - Multiple solutions to Nash game
  - Nonconvex objective or constraints
  - Existence of multipliers